XMUT204 Electronic Design

Demo 2 - Diode Circuit Simulation in LTSpice

1. Introduction

The SPICE circuit simulation programs provided in this demo are useful for modelling diodes in circuit simulations. The diode model is based on characterization of individual devices as described in a product data sheet and manufacturing process characteristics not listed. Some information have been extracted from a 1N4004 data sheet as shown in Figure 1 below.



Max avg rectified current I_O (A)	1	Forward voltage drop V_F (V)	1
Peak repetitive reverse voltage V_{RRM} (V)	400	@ I_F (A)	1
Peak forward surge current I_{FSM} (A)	30	Max reverse current I_R (μA)	5
Total capacitance C_T (pF)	15	@ V_R (V)	400

Figure 1: Partial manufacturer data sheet of 1N4004 diode.

The diode statement begins with a diode element name which must begin with "`d`" plus optional characters. Example diode element names include: `d1`, `d2`, `dtest`, `da`, `db`, `d101`.

Two node numbers specify the connection of the anode and cathode, respectively, to other components.

General form:

```
d[name] [anode] [cathode] [modelname]
.model ([modelname] d [parmtr1=x] [parmtr2=y] . . .)
```

Example 1:

```
d1 1 2 mod1
.model mod1 d
```

Example 2:

```
D2 1 2 Da1N4004
.model Da1N4004 D (IS=18.8n RS=0 BV=400 IBV=5.00u CJO=30 M=0.333 N=2)
```

The node numbers are followed by a model name, referring to a subsequent "`.model`" statement.

The model statement line begins with "`.model`", followed by the model name matching one or more diode statements.

Next, a "`d`" indicates a diode is being modeled.

The remainder of the model statement is a list of optional diode parameters of the form:

```
ParameterName=ParameterValue
```

None are used in Example 1 above. Example 2 has some parameters defined. For a list of diode parameters, see their details in the Table 1 in the following section.

The following sections outline the exercises of creating three models of diode in SPICE e.g. Default SPICE generic diode model, default 1N4004 model in SPICE (DI1N4004), and a custom model (Da1N4004) derived from manufacturer datasheet. We will then test these models in LTspice and follow by evaluating and comparing the outcomes of simulations of these three models.

2. SPICE Models for Diodes

Given in Table 1 is the default model of diode in SPICE. It is shown in the table most of parameters associated with a generic diode's characteristics and their relevant default values.

If you require specific diode model, the easiest approach to take for a SPICE model is the same as for a data sheet i.e. consult the manufacturer's web site. See also Table 2 in the following section for a list of the manufacturer's model parameters for some selected diodes.

A fall back strategy is to build a SPICE model from those parameters listed on the data sheet.

A third strategy, not considered here, is to take measurements of an actual device. Then, calculate, compare and adjust the SPICE parameters to the measurements.

Table 1: Default diode SPICE parameters

Symbol	Name	Parameter	Units	Default
IS	IS	Saturation current (diode equation)	A	1E-14
RS	RS	Parasitic resistance (series resistance)	Ω	0
n	N	Emission coefficient, 1 to 2	-	1
τD	TT	Transit time	s	0
CD(0)	CJO	Zero-bias junction capacitance	F	0
φ0	VJ	Junction potential	V	1
m	M	Junction grading coefficient:	-	0.5
–	–	• 0.33 for linearly graded junction	-	-
–	–	• 0.5 for abrupt junction	-	-
Eg	EG	Activation energy:	eV	1.11
–	–	• Si: 1.11	-	-
–	–	• Ge: 0.67	-	-
–	–	• Schottky: 0.69	-	-
pi	XTI	IS temperature exponent:	-	3.0
–	–	• pn junction: 3.0	-	-
–	–	• Schottky: 2.0	-	-
kf	KF	Flicker noise coefficient	-	0
af	AF	Flicker noise exponent	-	1
FC	FC	Forward bias depletion capacitance coefficient	-	0.5
BV	BV	Reverse breakdown voltage	V	∞
IBV	IBV	Reverse breakdown current	A	1E-3

If diode parameters are not specified as in "Example" model above, the parameters take on the default values listed in the table above. These defaults model refer to the integrated circuit type of the diodes. These values are certainly adequate for preliminary work with discrete devices.

For more critical and accurate work, use SPICE models supplied by the manufacturer, SPICE vendors, and other sources. Otherwise, derive some of the parameters from the data sheet.

3. Deriving the SPICE Models from Specification Sheets

There are several parameters of diode that are needed to create its model in SPICE. Some of these parameters can be determined from the manufacturer datasheet of the diode.

a. Emission Coefficient Parameter (N) – It controls forward current against voltage.
b. Saturation Current (IS) – It controls forward and reverse current against voltage.
c. Parasitic Resistance of Diode (RS) – It controls forward voltage at high current.
d. Transit Time (TT) – It controls switching reverse recovery characteristics.

e. Junction Capacitance (CJO) – It controls variation of capacitance with voltage.
f. Junction Grading Coefficient (M) – It controls variation of capacitance with voltage.
g. Junction Potential (VJ) – It controls variation of capacitance with voltage.
h. Activation Energy (EG) – It controls barrier height.
i. Temperature Exponent (XTI) – It controls variation in the operating temperature.
j. Reverse Breakdown Current (IBV) – It controls reverse breakdown characteristics.
k. Breakdown Voltage (BV) - It controls reverse breakdown characteristics.

There are also other SPICE parameters which are not covered in this exercise, such as: ISR and NR – these control reverse biased leakage, FC – it controls forward bias depletion capacitance coefficient, and so forth.

In the following sections, these parameters are discussed and analysed for their uses in the SPICE model.

3.1. Emission Coefficient Parameter (N)

First, select a value for diode's SPICE emission coefficient parameter, N, between 1 and 2. It is required for the diode equation (n). It is recommended that n, the emission coefficient is usually about 2. The value is used to estimate n in the Eber-Molls equation of a diode:

$$ID = IS(e^{VD/nVT} - 1)$$

The emission coefficient or ideality factor n accounts for the effect of recombination of holes with free electrons in the depletion region of the diode.

In the table below, we see that power rectifiers 1N3891 (12 A), and 10A04 (10 A) both use about 2. The first four in the table are not relevant because they are Schottky, Schottky, germanium, and silicon small signal, respectively.

Table 2: SPICE parameters for selected diodes (note: sk=Schottky; Ge=germanium; else silicon).

Part	IS	RS	N	TT	CJO	M	VJ	EG	XTI	BV	IBV
Default	1E-14	0	1	0	0	0.5	1	1.11	3	∞	1m
1N5711 sk	315n	2.8	2.03	1.44n	2.00p	0.333	-	0.69	2	70	10u
1N5712 sk	680p	12	1.003	50p	1.0p	0.5	0.6	0.69	2	20	-
1N34 Ge	200p	84m	2.19	144n	4.82p	0.333	0.75	0.67	-	60	15u
1N4148	35p	64m	1.24	5.0n	4.0p	0.285	0.6	-	-	75	-
1N3891	63n	9.6m	2	110n	114p	0.255	0.6	-	-	250	-
10A04 10A	844n	2.06m	2.06	4.32u	277p	0.333	-	-	-	400	10u
1N4004 1A	76.9n	42.2m	1.45	4.32u	39.8p	0.333	-	-	-	400	5u
1N4004 data sheet	18.8n	-	2	-	30p	0.333	-	-	-	400	5u

3.2. Saturation Current (IS)

The saturation current, IS, is derived from the diode equation, a value of (VD, ID) on the graph as given in the Figure 1 before. Then, N = 2 i.e. n in the diode equation (Eber-Molls equation) as shown below.

$$ID = IS(e^{VD/nVT} - 1)$$

From the theory, VT = 26 mV (at 25°C) and n = 2.0. Furthermore, we obtain that VD = 0.925 V at 1 A from graph in Figure 1.



As a result, by entering all of these values, the saturation current of the diode is found to be:

$$1\ A = IS(e^{(0.925\ V)/(2)(26\ mV)} - 1)$$

$$IS = 18.8E\text{-}9$$

The numerical values of IS = 18.8 nA and N = 2 are entered in last line of the table above for comparison to the manufacturers model for 1N4004, which is considerably different.

3.3. Parasitic Resistance of Diode (RS)

Parasitic series resistance of the diode, RS is assigned to default value i.e. 0 Ω for now and it will be estimated later. For now, the most important parameters in the DC static of the diode are N, IS, and RS.

That resistance represents the behavior of the interface between the diode semiconductor material and the metal contact used for the terminal of the diode. Unlike resistors, though, diodes are not linear devices. This means that the resistance of diodes does not vary directly and proportional to the amount of voltage and current applied to them.

An actual diode offers a very small resistance (not zero) when forward biased and is called a forward resistance.

Some other parasitic resistance present due to impedance due to parasitic capacitance and inductance in the diode.

3.4. Transit Time (TT)

When a forward-biased diode has a reverse voltage applied across it, it takes time for the charge to dissipate and hence for the diode to turn off. The time taken for the diode to turn off is captured primarily by the transit time parameter. Once the diode is off, any remaining charge then dissipates, the rate at which this happens being determined by the carrier lifetime.

It is suggested that the transit time, TT or τD, be approximated from the reverse recovery stored charge QRR, a data sheet parameter (i.e. not available in our data sheet) and IF, forward current (e.g. see the following equation).

$$ID = IS(e^{VD/nVT} - 1) \quad \tau D = QRR/IF$$

We take the TT = 0 second default for lack of QRR. Although, it would be reasonable to take TT for a similar rectifier like the 10A04 at 4.32 us. In the table given above, the 1N3891's TT is not a valid choice because it is a fast recovery rectifier.

3.5. Junction Capacitance (CJO)

This parameter is about the parasitic capacitance that exist in the diode. CJO, the zero bias junction capacitance is estimated from the VR vs CJ graph as shown in Figure 1. In the graph, the capacitance at the nearest to zero voltage on the graph is 30 pF at 1 V.



If simulating high-speed transient response, as in switching regulator power supplies, TT and CJO parameters must be provided.

3.6. Junction Grading Coefficient (M)

The junction grading coefficient M is related to the doping profile of the junction. This is not a data sheet item.

The default is 0.5 for an abrupt junction. We opt for M = 0.333 corresponding to a linearly graded junction. The power rectifiers in the table given above use lower values for M than 0.5.

3.7. Junction Potential (VJ) and Activation Energy (EG)

Junction potential is the voltage drop across the depletion region of the PN area of the diode.

EG is the energy gap for the semiconductor type measured in Joules (J). The value for silicon is usually taken to be 1.11 eV, where 1 eV is 1.602e-19.

We take the default values for the junction potential, VJ and activation energy, EG. Many diodes use VJ = 0.6 than shown in the table above. However, the 10A04 rectifier uses the default, which we use for our 1N4004 model (Da1N4001 in the Table 2 above). Use the default EG = 1.11 for silicon diodes and rectifiers. The table above also lists values for Schottky and germanium diodes.

3.8. Temperature Exponent (XTI)

XTI is the saturation current temperature exponent. This is usually set to 3.0 for pn-junction diodes, and 2.0 for Schottky barrier diodes.

Take the IS temperature exponent, XTI = 3, in this case it is basically the default IS temperature coefficient for silicon devices. See the table above for XTI for Schottky diodes.

Appropriate values for *XTI* and *EG* depend on the type of diode and the semiconductor material used. Default values for particular material types and diode types capture approximate behavior with temperature. The block provides default values for common types of diode.

In practice, the values of *XTI* and *EG* need tuning to model the exact behaviour of a particular diode. Some manufacturers quote these tuned values in a SPICE Netlist, and you can read off the appropriate values. Otherwise, you can determine improved estimates for *EG* by using a datasheet-defined current-voltage data point at a higher temperature. The block provides a parameterization option for this. It also gives the option of specifying the saturation current at a higher temperature IS_{Tm2} directly.

You can also tune the values of *XTI* and *EG* yourself, to match lab data for your particular device.

3.9. Reverse Breakdown Current (IBV) and Breakdown Voltage (BV)

These parameters refer to the current (IBV) and voltage (BV) whenever diode is operated in the reverse bias mode.



Max avg rectified current I_O (A)	1	Forward voltage drop V_F (V)	1
Peak repetitive reverse voltage V_{RRM} (V)	400	@ I_F (A)	1
Peak forward surge current I_{FSM} (A)	30	Max reverse current I_R (µA)	5
Total capacitance C_T (pF)	15	@ V_R (V)	400

In the abbreviated data sheet, as shown in the figure above, lists IR = 5 µA @ VR = 400 V, corresponding to IBV = 5 uA and BV = 400 V respectively.

The 1N4004 SPICE parameters derived from the data sheet are listed in the last line of the table above for comparison to the manufacturer's model listed above it.

BV is only necessary if the simulation exceeds the reverse breakdown voltage of the diode, as is the case for zener diodes. IBV, reverse breakdown current, is frequently omitted, but may be entered if provided with BV.

4. Comparing Diode Models from Different Sources

4.1. First Trial

Figure below shows a circuit to compare the manufacturers model, the model derived from the datasheet, and the default model using default parameters. The three dummy 0 V sources are necessary for diode current measurement. The 1 V source is swept from 0 to 1.4 V in 0.2 mV steps (e.g. see `.DC` statement in the netlist given below). DI1N4004 is the manufacturer's diode model, Da1N4004 is our derived diode model.



Figure 2: SPICE circuit for comparison of: manufacturer model (D1), calculated datasheet model (D2), and default model (D3).

In your LTspice software, create a new schematic.

Click on the directive button in the toolbar and type the SPICE directive as shown below. Place and click the directive somewhere in the schematic.

```
.DC V1 0 1400mV 0.2m
.model DI1N4004 D (IS=76.9n RS=42.0m BV=400 IBV=5.00u CJO=39.8p
+M=0.333 N=1.45 TT=4.32u)
.model Da1N4004 D (IS=18.8n RS=0 BV=400 IBV=5.00u CJO=30p
+M=0.333 N=2.0 TT=0)
.model Default D
```

Notice that in the directive, the first directive is to step the voltage source V1 from 0 to 5 V in the increment of 0.2 mV and the rest of the directive is for the SPICE netlist parameters: (D1) DI1N4004 manufacturer's model, (D2) Da1N40004 datasheet derived, (D3) default diode model.

Add a voltage component as V1 and assign it to be 1 V DC.

Add three other voltage sources to the schematic and assign each of them as 0 V.

Add three diodes to the schematic.

Press CTRL +right click on top of the first diode component to assign D1 to be DI1N4004. Then perform the same steps to assign D2 to be Da1N4004 and let D3 to be D.

Add a ground component also to the schematic.

Click on the simulation button.

Add the plots to the viewing window related to the currents flowing in three diodes in the schematic.

If the plot is uncharted, change the x-axis of each plot to be logarithmic.

Finally, we compare the three models in the figures shown below and with the graph data in the datasheet.



Figure 3: First trial of manufacturer model, calculated datasheet model, and default model.

By referring to the plots and table given below, comparison of the forward bias currents characteristics of the three model show that the default model is good at low currents, the manufacturer's model is good at high currents, and our calculated datasheet model is best of all up to 1 A.

Agreement is almost perfect at 1 A because the IS calculation is estimated based on diode voltage at 1 A. But, in the end our model **grossly** over states current above 1 A as shown in the plots shown above.

By referring to the table below (e.g. values taken from the simulation plots), VD is the diode voltage versus the diode currents for the manufacturer's model, our calculated datasheet model and the default diode model e.g. the last column “1N4004 graph” is from the datasheet voltage versus current curve given in the Figure 1 which we attempt to match.

Table 3: Comparison of manufacturer model, calculated datasheet model, and default model to 1N4004 datasheet graph of V vs I (1st trial).

#	VD	Manufacturer model	Datasheet model	Default model	1N4004 graph
1	0.700	1.612924	0.01416211	0.005674683	0.01
2	0.800	3.346832	0.09825960	0.2731709	0.13
3	0.900	5.310740	0.6764928	12.94824	0.70
4	0.925	5.823654	1.096870	34.04037	1.00
5	1.000	7.395953	4.675526	618.5078	2.00
6	1.100	9.548779	30.231452	2,9544.71	3.30
7	1.200	11.74489	223.3392	1,411,283	5.30
8	1.300	13.97087	1,543.591	67,413,790	8.00
9	1.400	16.21861	10,668.40	3,220,203,000	12.0

4.2. Second Trial

To improve the performance of our diode model, the solution is to increase RS from the default RS = 0 Ω. Changing RS from 0 Ω to 8 mΩ in the datasheet model causes the curve to intersect 10 A (not shown) at the same voltage as the manufacturer’s model.



Increasing RS to 28.6 mΩ shifts the curve further to the right as shown in the Figure 4 below. This has the effect of more closely matching our datasheet model to the datasheet graph.

```
.model Da1N4004 D (IS=18.8n RS=28.6m BV=400 IBV=5.00u CJO=30p
+M=0.333 N=2.0 TT=0)
```



Figure 4: Second trial to improve calculated datasheet model compared with manufacturer model and default model.

Table below shows that the current 12.24470 A at 1.4 V matches the graph at 12 A. However, the current at 0.925 V has degraded from 1.096870 A above to 0.7318536 A. Changing Da1N4004 model statement RS = 0 Ω to RS = 28.6 mΩ decreases the current at VD = 1.4 V to around 12.2 A.

Table 3: Comparison of manufacturer model, calculated datasheet model, and default model to 1N4004 datasheet graph of V vs I (2^{nd} trial).

#	VD	Manufacturer model	Datasheet model	1N4001 graph
1	0.701	1.628276	0.01432463	0.01
2	0.800	3.343072	0.09297594	0.13
3	0.900	5.310740	0.5102139	0.70
4	0.925	5.823654	0.7318536	1.00
5	1.000	7.395953	1.763520	2.00

6	1.100	9.548779	3.848553	3.30
7	1.200	11.74489	6.419621	5.30
8	1.300	13.97087	9.254581	8.00
9	1.400	16.21861	12.24470	12.00

Further improvement: decrease N so that the current at VD = 0.925 V is restored to 1 A. This may increase the current (12.2 A) at VD = 1.4 V requiring an increase of RS to decrease current to 12 A.

5. Zener Diode

There are two approaches to modelling a zener diode: set the BV parameter to the zener voltage in the model statement, or model the zener with a subcircuit containing a diode clamper set to the zener voltage.

An example of the first approach sets the breakdown voltage (BV) of the zener diode to 15 V for the 1N4469 15 V zener diode model (IBV optional):

```
.model D1N4469 D ( BV=15 IBV=17m )
```

Creating the circuit based on this model in LTspice



The circuit above will give you a plot of reverse bias of I-V curve graph of the given zener diode in the viewing window that looks like shown below (note: change the x-axis plot to be logarithmic if necessary).



The second approach models the zener with a subcircuit. Clamper D1 and VZ in Figure below models the 15 V reverse breakdown voltage of a 1N4477A zener diode. Diode DR accounts for the forward conduction of the zener in the subcircuit.



Figure 5: Zener diode subcircuit uses clamper (D1 and VZ) to model zener.

In the LTspice, create a new schematic.

For the zener diode given above, type the following SPICE directive for the subcircuit.

```
.SUBCKT DI-1N4744A 1 2
* Terminals A K
D1 1 2 DF
DZ 3 1 DR
VZ 2 3 13.7
.MODEL DF D ( IS=27.5p RS=0.620 N=1.10
+CJO=78.3p VJ=1.00 M=0.330 TT=50.1n )
.MODEL DR D ( IS=5.49f RS=0.804 N=1.77 )
.END
```

Add an instance of the diode symbol to your schematic.

Move the cursor over the body of the diode symbol and Ctrl + Right-Click. A dialog box appears.

Change Prefix: “D” to “X”. The symbol now netlists as a subcircuit instead of a generic diode.

Change “D” in the Value to be “DI-1N4744A”, corresponding to the name on the .SUBCKT line.



Click OK.

[Note: this assumes the third-party model we are adding follows popular pin order conventions].

Run a DC sweep simulation from 0 until 12 V with 0.2 mV increment



Obtain a plot of the current that flows in the zener diode into the viewing window (note: change the x-axis plot to be logarithmic if necessary).



6. Other Types of Diode

You might see also that there are other types of diodes in the market. Model of these other diodes require different modelling in the SPICE.

Tunnel diode: A tunnel diode may be modeled by a pair of field effect transistors (JFET) in a SPICE subcircuit. This diode is typically used in the oscillator circuits.

Gunn diode: A Gunn diode may also be modeled by a pair of JFET's. This diode is used for microwave relaxation oscillator applications.

Review Summary

a. Diodes are described in SPICE by a diode component statement referring to `.model` statement. The `.model` statement contains parameters describing the diode. If parameters are not provided, the model takes on default values.

b. Static DC parameters include N, IS, and RS. Reverse breakdown parameters: BV, IBV.

c. Accurate dynamic timing requires TT and CJO parameters.

d. Models provided by the manufacturer are highly recommended.